Параллельные процессы в RSL
RSL предоставляет возможность специфицировать параллельные процессы. Точнее, предоставляет комбинаторы для спецификации параллельных вычислений выражений. Более того, выражения могут взаимодействовать между собой посредством каналов.

Следующий модуль определяет один буфер , opb, который взаимодействует с окружением через два канала add и get. Значения типа Elem вводятся из канала add и выводятся в get.

ONE_PLACE_BUFFER =

 class 

   type Elem

   channel add, get : Elem

   value opb : Unit -> in add out get Unit
   axiom opb() is let v = add? in get!v end; opb()

 end
Объявление каналов.

Объявление выглядит так 


channel



channel_definition1

channel_definitionN


для n>=1.

сhannel_definition имеет вид 


id1, … ,idN : type_expr

Каналы id1, … idN передают значения типа type_expr.

Функции с доступом к каналам.

Функция opb в примере имеет тип:

Unit -> in add out get Unit

Это функция которая, при вызове, взаимодействует с окружением через каналы add и get. Точнее, получает значения из окружения через канал add, и посылает значения через get. Функции с доступом к каналам, такие как opb также называют процесс.
Выражения взаимодействия.

RSL предоставляет два примитива взаимодействия: ввод значения из канала и вывод значения в канал. В области видимости объявления:


channel id : T

в выражении можно специфицировать ввод из канала выражением ввода вида:


id?

Вычисление выражения приостанавливается, до вывода значения в канал id другим процессом. После ввода из канала id, принятое значение возвращается как результат вычисления выражения ввода. Тип выражения ввода совпадает с типом значений канала.

В выражении можно специфицировать вывод значения в канал выражением вывода вида:


id!value_expr

выражение value_expr, вычисляется, и результат выводится в канал id. Вычисление выражения приостанавливается, до ввода значения в канал другим процессом. Тип value_expr должен совпадать с типом канала. Тип всего выражения вывода Unit.

Наш пример содержит выражение ввода add? и выражение вывода get!v. 
Таким образом процесс opb многократно вводит значение из канала add и выводит в канал get.

Составление параллельных выражений.

Взаимодествие через каналы – это средство взаимодействия выражений, вычисляющихся параллельно. Два выражения могут вычисляться одновременно, как следующие:


value_expr1 || value_expr2

Два выражения вычисляются параллельно, пока одно из них не завершится, после чего продолжается  вычисление другого. Оба выражения должны иметь тип Unit, который имеет и всё выражение.

В качестве примера рассмотрим следующие определения:


channel c : Int

variable x : Int
В области видимости этих выражений, выражения x:=c? и c!5 могут вычислятся параллельно:


x:=c? || c!5
Вычисление этих выражений может привести к взаимодействию, так как правое выражение выводит значение 5 в канал, которое читается левым выражением. Если взаимодействие происходит то эффект от вычисления этих выражений следующий:


x:=5

Взаимодействие синхронизированно: выражение вывода специфицирует вывод в канал только если выражение ввода одновременно специфицирует ввод из канала. Параллельные попытки записи в канал и чтения из канала необязательно приводят к взаимодействию. Что произойдёт зависит от внутреннего выбора. Таким образом два выражения  могут параллельно взаимодействовать с третьим выражением. Например:


(x:=c? || c!5) || c!7

один из возможных эффектов:


x:=7; c!5

То есть, правое выражение выводит значение 7 в канал c. Левое выражение читает значение и сохраняет его в x. После взаимодействия остается невыполненным c!5.

Заметим, однако, что эффектом может быть:


x:=5; c!7

Либо, даже, взаимодействие может не происходить вообще.

Комбинатор парраллельной композиции || коммутативен и ассоциативен.

Для примера предположим что мы хотим использовать один буфер для соединения процессов reader и writer. Процесс reader вводит значения из канала input и процесс writer выводит значения в канал output. Значения от процесса reader передаются процессу writer через буфер opb.

READER_WRITER


extend ONE_PLACE_BUFFER with

class


 type Input, Output


 channel input : Input, output : Output


 value 



Transform1 : Input -> Elem,



Transform2 : Elem -> Output,



Reader : Unit -> in input out add Unit,



Writer : Unit -> in get out output Unit

 axiom



Reader() is let v = input? in add!(transform1(v))end; reader(),



Writer() is let v = get? in output!(transform2(v)) end; writer()


end
Процесс reader многократно вводит значение v из канала input и выводит значение transform1(v) в канал add. Процесс writer многократно вводит значение v из канала get и выводит значение transform2(v) в канал output.

 Запустим эти процессы reader, opb, writer параллельно.

SYSTEM = 

 extend READER_WRITER with
 class


value system : Unit ->in input, add, get out output, add, get Unit


axiom system() is reader() || opb() || writer()

 end

Внешний выбор.

Вернёмся к определению аксиомы буфера opb:

   axiom opb() is let v = add? in get!v end; opb()

Процесс opb после начала работы предлагает один тип взаимодействия с окружением: ввод из канала add. После ввода предлагается другой тип взаимодействия: вывод в канал get. 

 Однако бывают ситуации в которых мы хотим чтобы процесс предлагал одновременно несколько типов взаимодействий.

Комбинатор внешнего выбора |=| служит для спецификации выбора между различными типами взаимодействий. В качестве примера, рассмотрим следующее определение:


channel c,d : Int


variable x : Int
Рассмотрим выражение внешнего выбора


x:=c? |=| d!5
Данное выражение предлагает два взаимодействия: либо ввод из канала c, либо вывод в канал d. Выбор назавается внешним, так как его проиводит окружение (т.е. другие выражения выполняющиеся одновременно с этим). Предположим мы запустили это выражение параллельно с выражением c!1

(x:=c? |=| d!5) || c!1

Возможный эффект этого выражения состоит в том, что значение 1 передаётся через канал c:


x:=1

Синтаксис:


value_expr1 |=| value_expr2

Оба выражения дожны иметь одинаковый тип, который также является типом всего выражения. Обычно value_expr1 и value_expr2 начинаются с некоторого типа взаимодействия. Только одно из выражений будет вычисленно, в зависимости от того какой тип взаимодействий хочет произвести окружение.

Комбинатор внешнего выбора ассоциативен и коммутативен.

Внутренний выбор.

Вспомним, что комбинатор внешнего выбора выражает выбор между двумя выражениями. Термин “внешний” значит что окружение может влиять на выбор выражения. Рассмотрим пример:


(x:= c? |=| d!5) || c!1
Из x:=c? и d!5 выбирается x:=c?,  из-за события c!1.

В дополнение к комбинатору внешнего выбора RSL предоставляет комбинатор внутреннего выбора |^|, который специфицирует внутренний выбор между выражениями:


value_expr1 |^| value_expr2

Выбор value_expr1 или value_expr2 зависит от внутреннего выбора, на который не может повлиять окружение. Оба выражения должны иметь одинаковый тип, который является типом всего выражения.

Комбинатор внутреннего выбора ассоциативен и коммутативен. Рассмотрим следующее выражение:


(x:=c? |^| d!5) || c!1

Выражение c!1 не влияет на то какое из выражений x:=c? или d!5 выбирается. Если внутренний выбор выпадает на x:=c?, то результат - выражение:


x:= c? || c!1
потенциально приводящее к взаимодействию через канал c. Если выбор падает на d!5, то результат - выражение:


d!5 || c!1

препятствующее внутренним взаимодействиям.

Interlock комбинатор.

Рассмотрим алгебраическую спецификацию списка:

LIST_A =

 class


type


 List


value


 empty_a : List,


 add_a : Int >< List -> List,


 head_a : List -~-> Int,


 tail_a : List -~-> List


axiom all i : Int, l : List :-


 [head_add]



head_a(add_a(i,l)) = i,


 [tail_add]



tail_a(add_a(i,l)) = l

 end

теперь рассмотрим спецификацию этого списка в виде процесса:

LIST = 

 extend LIST_A with
 class 

  channel

 
is_empty : Bool,


add, head : Int,


tail : Unit
  value 


list : List -> in add, tail out is_empty, head Unit
  axiom all l : List :-


list(l) is


is_empty!(l = empty_a); list(l)



|=|



let i = add? in list(add_a(i,l)) end


|=|



if ~(l = empty_a) 




then head!(head_a(l))); list(l)




else stop


end



|=|



if~(l = empty_a) 




then tail?; list(tail_a(l)) 




else stop



end

 end

Процесс list взаимодействует с окружением через каналы is_empty, add, head, tail. Процесс характеризуется состоянием ‘state’ имеющим тип List (параметр процесса). Описание процесса дано через функции модуля LIST_A.

Попробуем специфицировать процесс список в стиле алгебраических спецификаций. Рассмотрим аксиому head_add:


 [head_add]



head_a(add_a(i,l)) = I

Нам потребуется написать чуть больше. Определим переменную head_res для хранения значения возвращаемого по каналу head.

variable head_res : Int

Теперь можно попытаться написать аксиому так:

axiom all i : Int, l : List :-


[head_add]


 list(l) || (add!i; head_res := head?) is


list(l) || (add!i; head_res := i)
Так как комбинатор || не отменяет параллельные взаимодействия, то между записью add!i в канал и head_res := i, возможна запись add!i1,т.е также должна выполнятся аксиома:

all i,i1 : Int, l : List :-


add!i1 || (list(l) || (add!i; head_res := head?)) is


add!i1 || (list(l) || (add!i; head_res := i))

может произойти следующее:

1. list(l) взаимодействует с add!I:
add!i1 || list(add_a(i, l)) || head_res := head?

2. list(add_a(i, l)) взаимодействует с add!i1:
list(add_a(i1, add_a(i, l))) || head_res := head?

3. list(add_a(i1, add_a(i, l))) взаимодействует с head?
head_res := i1; list(add_a(i1, add_a(i, l))) 

что очевидно не верно.

Решение состоит в использовании комбинатора ++ (interlock), который более «агрессивен» чем комбинатор ||. 
Interlock комбинатор имеет вид:

value_expr1 ++ value_expr2

Выражение вычисляется  вычислением компонент value_expr1, value_expr2: компоненты вычисляются параллельно, пока одно из них не завершится, далее продолжается вычисление другого (как для комбинатора ||). Однако, в течении параллельного вычисления не происходит взаимодействий с окружением (после того как один закончился, другой может взаимодействовать с окружением). В нашем примере внешнеее взаимодействие с add!i1 не происходит.

Примеры применения комбинатора interlock (++):

value e,e1,e2 : T

channel c,c1,c2 : T

variable x : T

x:=c? ++ c!e is x:=e

Т.е. так как выражения x:=c? и c!e могут взаимодействовать, то они будут взаимодействовать.

При использовании параллельного комбинатора взаимодействия более сложные:

x:=c? || c!e is (x:=e) |^| ((x:= c?; c!e)|=|(c!e; x:=c?)|=|(x:=e))
В зависимости от внутреннего выбора:

либо x:=e – если выполняется как interlock, 

либо вычисление завит от внешнего выбора.

(x:=c1? |=| c2!e2) ++ c1!e1 is x:=e1
interlock комбинатор выбирает выражение x:=c1?.

При внутреннем выборе может произойти останов: 

(x:=c1? |^| c2!e2) ++ c1!e1 is x:=e1 |^| stop

Пример показывающий как невозможность взаимодействий с окружением приводит к останову:

x:=c1? ++ c2!e is stop
в отличии от параллельного комбинатора:

x:=c1? || c2!e is (x:=c1?; c2!e) |=| (c2!e; x:=c1?)

Правильная аксиома head_add:

axiom all i : Int, l : List :-


[head_add]


 (list(l)++add!i) ++ head_res := head? is


(list(l)++add!i) ++ head_res := i

